

Amstrad moniteur

```

10 REM #####
20 REM ##
30 REM ##      moniteur de code machine      ##
40 REM ##      copyright E.charton et amstrad magazine ##
45 REM ##      amsmoniteur                    ##
50 REM ##
60 REM #####
70 REM #####
80 REM ##
90 REM ##      initialisation                  ##
100 REM #
110 REM #####
115 MODE 2
120 WINDOW #1,5,62,4,14:PEN#1,0:PAPER#1,1:CLS#1
130 WINDOW #2,5,50,16,17
140 WINDOW #3,3,68,19,23:PEN#3,0:PAPER#3,1:CLS#3
150 LOCATE 22,1:PRINT " -----moniteur de code machine-----"
155 ON BREAK GOSUB 20000
160 GOTO 10000
200 REM #####
210 REM ##
220 REM ##      examen memoire                  ##
230 REM ##
240 REM #####
250 INPUT#2,"adresse de debut",adr
260 CLS #1:PRINT#1,"----debut---> ";adr:PRINT
270 adr=ABS(adr):PRINT#1,adr;" /h";HEX$(adr);"----> ";PEEK(adr);" / h ";HEX$(PEEK (adr));" / b ";
BIN$(PEEK(adr));" /chr# ";
275 IF PEEK (adr)>32 THEN PRINT#1, CHR$(PEEK(adr));
276 PRINT#1
280 adr=ABS(adr)+1
285 IF INKEY$="m" OR INKEY$="M" THEN GOTO 10000
290 GOTO 270
3000 REM #####
3010 REM ##
3020 REM ##      modification de la memoire      ##
3030 REM ##
3040 REM #####
3045 CLS#1
3050 INPUT #2,"adresse de depart a modifier? ",adr
3055 IF adr >65535 THEN 3050
3060 PRINT #1,adr;" /h ";HEX$(adr);" -->";
3080 INPUT #1,al$

```

Nous allons vous permettre ce mois-ci d'examiner plus en profondeur la mémoire de votre Amstrad grâce au programme Amstrad moniteur.

Ce moniteur de code machine vous donne des relevés du contenu des cases mémoires sous diverses formes qui sont :

Adresse en décimal (de 0 à 65535).

Adresse en hexadécimal (de 0 à FFFF).

Contenu en décimal (0 à 255).

Contenu en hexadécimal (0 à FF).

Contenu en binaire (ex : 00001001).

Contenu sous forme de caractère ASCII.

Plusieurs options concernant l'examen ou la modification de la mémoire vous sont offertes.

L'option 1 concerne l'examen, elle vous indique dans la fenêtre du haut de l'écran le contenu de la mémoire de la manière suivante : premièrement l'adresse en décimal puis en hexadécimal (précédée par la lettre h minuscule). Puis le code machine en décimal, hexadécimal, binaire (précédé par b), caractère graphique (précédé par CHRS), il est important de noter pour le caractère graphique que si le code est inférieur à 33, il s'affi-

```

3081 IF a1$="" THEN 3095
3082 IF a1$="m" OR a1$="M" THEN 10000
3085 IF VAL(a1$)>255 THEN 3060
3090 POKE (adr),VAL (AL$)
3095 adr=adr+1
3100 GOTO 3060

10000 REM #####
10010 REM ## ##
10020 REM ## menu ##
10030 REM ## ##
10040 REM #####
10045 CLS#3
10050 PRINT#3," -----menu----- "
10060 PRINT#3," 1--->examen memoire"
10070 PRINT#3," 2--->modification de la memoire
10080 PRINT#3," 3--->retour au basic
10090 PRINT#3," 4--->examen memoire et sortie sur listing"
10100 REM ##### saisie du choix #####
10110 INPUT#2 ,"menu:votre choix?",a$
10120 IF a$="1" THEN 200
10130 IF a$="2" THEN 3000
10140 IF a$="3" THEN MODE 2:STOP
10150 IF a$="4" THEN GOTO 30000
10300 GOTO 10110

20000 REM #####
20010 REM ## ##
20020 REM ## on break ##
20030 REM ## ##
20040 REM #####
20050 LOCATE 2,24:PRINT "pause"
20060 IF INKEY$="" THEN 20060
20070 LOCATE 2,24:PRINT " "
20080 RETURN

30000 REM #####
30010 REM ## ##
30020 REM ## examen memoire ##
30040 REM ## et sortie sur listing ##
30045 REM ## ##
30050 REM #####
30250 INPUT#2,"adresse de debut",adr
30260 CLS #1:PRINT#1,"----debut---> ";adr:PRINT
30261 CLS #1:PRINT#8,"----debut---> ";adr
30270 adr=ABS(adr):PRINT#1,adr;" /h";HEX$(adr);"---> ";PEEK(adr);" / h ";HEX$(PEEK (adr));" / b
";BIN$(PEEK(adr));" /chr$ ";
30271 PRINT#8,adr;" /h";HEX$(adr);"---> ";PEEK(adr);" / h ";HEX$(PEEK (adr));" / b ";BIN$(PEEK(a
dr));" /chr$ ";
30276 PRINT#8:PRINT#1:adr=ABS(adr)+1:IF INKEY$="m" OR INKEY$="M" THEN GOTO 10000
30290 GOTO 30270

```

che un espace derrière CHR\$, puisque les valeurs comprises entre 0 et 32 correspondent à des caractères de contrôle (voir pour cela le manuel Amstrad, page A3.1). Pour sélectionner l'option 1 tapez 1 lorsque vous êtes au menu, l'ordinateur vous demande l'adresse de départ entrez alors une adresse en 0 et 65535 décimale, ou entre 0 et FFFF, pour une valeur hexadécimale, (n'oubliez pas de la faire précéder par le code & afin de stipuler qu'il s'agit d'un chiffre hexadécimal : ex &7FFF). Pour interrompre le déroulement de l'affichage, à tout moment, il suffit d'appuyer sur « ESC » et de rappuyer pour faire repartir le déroulement. Pour revenir au menu il suffit d'appuyer sur la touche M.

L'option 2 vous permet de modifier le contenu de la mémoire, par exemple pour entrer des tables de données, éventuellement faire de courtes modifications ou bien entrer des programmes machines de faible importance.

Du menu, tapez 2 afin d'accéder à l'option, puis à la question concernant l'adresse de départ, entrez l'adresse à laquelle doivent commencer les modifications.

Vous allez voir apparaître à l'écran l'adresse en décimal et hexadécimal, suivie d'un point d'interrogation. Trois possibilités s'offrent à vous, à ce moment : vous pouvez entrer une valeur comprise entre 0 et 255 suivie de ENTER, appuyer sur ENTER directement, ce qui ne modifiera pas la mémoire ou taper « m » pour revenir au menu.

L'option 3 vous permet le retour au Basic par exemple pour tester un programme ou effectuer une sauvegarde.

L'option 4 est dédiée aux possesseurs d'imprimante, elle possède toutes les caractéristiques de l'option 1, mais effectuée en plus une sortie des informations sur imprimante.

E. Charton